The Effect of Extended Unemployment Insurance Benefits: Evidence from the 2012-2013 Phase-Out

A Replication Study of Farber, Rothstein, and Valletta (2015)*

Nikhil SEBASTIAN†

November 14, 2021

Abstract

This essay presents a replication study of Farber, Rothstein, and Valletta (2015) to confirm the robustness and validity of its results. The replication exercise suggests that the results are valid and robust in general but, the validity of the estimated results of one specification of the model is questionable as there seem to be some observable differences between the estimated results from the main analysis and the replication study. To complement this study, I provide scripts that enable replicators to access and automate the download of additional data required to replicate the paper.

* The replication materials that I use are maintained in a remote repository and can be accessed at https://github.com/nikhilsebastiank/FRV2015-ReplicationMaterials

†Email address: nikhil.sebastian@sciencespo.fr

1 Introduction

The COVID-19 pandemic caused a substantial increase in the level of unemployment, thereby making labor markets weak, especially during the year 2020. Even though the level of unemployment has declined over the period of the pandemic, it still remains higher than pre-pandemic levels (Smith, Edwards, and Duong (2021)). This trend is similar to the case of the Great Depression that was observed between the years 2007 and 2009. In order to respond to declining labor-market conditions, policy-makers in the U.S. responded by extending the duration of unemployment insurance benefits from a regular period of about 26 weeks up to a maximum of 99 weeks in some states in the during the Great Recession (Rothstein (2011)). The problem of moral hazard arising from asymmetric information about the actions of unemployed workers makes the determination of optimal level of benefits difficult. Additionally, when one thinks about optimal unemployment insurance benefits, it is essential to take into account various dimensions like eligibility conditions, amount of compensation and duration of the benefits that are offered to those who are eligible. There is varied opinion among researchers and policy-makers about the optimal level of insurance precisely due to the moral hazard problem.

Farber, Rothstein, and Valletta (2015) seek to close some gaps in the literature regarding the optimal level of unemployment insurance through their analysis of the extension of unemployment insurance benefits during the Great Recession. They contribute to existing literature in two significant ways. First, they use a novel methodology to simulate benefit duration and match micro-level Current Population Survey data to estimate the effect of extended unemployment insurance benefits on exit from measured unemployment. Second, they identify the channels through which this effect is propagated. Their findings suggest that there is a negative effect on exit from unemployment, but this effect is primarily driven through the channel of labor force attachment, or in other words, the incentive of an unemployed worker to stay in the labor force to be eligible to claim the benefits.

In this essay, I present a replication exercise conducted to confirm the robustness of the results presented in their main analysis. This essay organized as follows: Section 2 presents a summary of the paper where I discuss the main research questions that the authors seek to address, some background that sheds light on existing literature, data and methodology adopted in the analysis and conclude by presenting the findings. Section 3, presents the replication exercise and the check the robustness of the results. Section 4 concludes with a discussion of the key findings, some limitations of the replication exercise and additional interesting questions. To complement this essay, I also provide some R and Julia scripts that I use for the replication which is maintained on a remote repository. The code used for the replication and the reference figures and table from the main analysis are provided in the appendix.

2 Summary of the Paper

In this section, I summarize the paper (Farber, Rothstein, and Valletta (2015)) as follows: First, I motivate the choice of their research question and provide background that is useful to

understand the context of the main results. Then, I describe the data and methodology that the authors have used and conclude by presenting the main results and potential explanations for these results.

2.1 Research Question

The research question that the authors seek to address is whether the extension of unemployment insurance benefits has an effect on unemployment and if yes, how? The complex question of optimal unemployment insurance benefits is far from being answered perfectly: owing to its multi-dimensionality and the moral hazard problem arising from asymmetric information about the actions of the unemployed individuals. The authors build up on this literature by analyzing the effect of UI[1] extension on unemployment over the expansion-contraction phase of the extension period. In order to counter the serious effects of the Great Recession on labor markets, policy makers in the United States responded by extending the duration of unemployment benefits. This policy response to the Great Recession in the U.S. provided the authors a quasi-natural experiment to analyze the effect of the extension of unemployment insurance benefits on unemployment and to understand through what channels it affected unemployment. Further, additional data about the level of unemployment after the phase-out of the extensions was used to update their previous analyses that were done independently in Rothstein (2011) and Farber and Valletta (2015). The authors wanted to investigate whether the persistence in weak labor market conditions during the recovery could be explained by UI benefits extensions, as this would have policy implications for economic efficiency.

2.2 Background

The analysis of the impact of unemployment insurance benefits on unemployment has been of key interest in the field of labor-economics for a long time. Using a job search framework, the seminal paper by Baily (1978) predicts that the duration of unemployment is positively related to the level of unemployment benefits and provides a formula to capture the relationship between the optimal level of UI benefits, risk aversion and the elasticity of unemployment duration with respect to unemployment benefits. Further, it has been shown by Chetty (2006) that formula still holds in general environments (Cahuc, Carcillo, and Zylberberg (2014)).

More recently, Rothstein (2011) and Farber and Valletta (2015), in simultaneous[2] but independent studies show the effect of unemployment insurance benefits and its extension in the U.S. during the Great Recession on unemployment. Both these papers focus on the behaviour of workers using a job search framework but use different approaches to identify the impact of unemployment insurance benefits and its extension on measured unemployment. For instance, Rothstein (2011), using CPS[3] data identifies causal effects of UI on unemployment induced by variations in benefit duration (through the timing of policies),

[1] Unemployment Insurance

[2] With reference to the period during which the studies were undertaken.

[3] Current Population Survey

UI-eligibility, state decision to accept or decline EB[4] provisions, and differences in remaining UI duration for eligible workers. He identifies two effects: the labor force exit effect and the re-employment effect, through which the unemployment insurance benefits and its extension impacts measured unemployment. The labor force exit effect refers to the channel through which UI extension incentivizes unemployed workers to continue to be in the labor force to be eligible for benefits. The re-employment effect refers to the channel through which UI extension incentivizes unemployed workers to look for jobs, thereby increasing their probability of re-employment. His findings predict that the UI extension was associated with a higher rate of unemployment, but much of this impact was through the channel of labor force attachment, or in other words, due to the labor force exit effect. These results are consistent with the findings of Farber and Valletta (2015).

2.3 Data and Methodology

The authors have used data on the EB program and the EUC[5] program obtained from the U.S. Department of Labor (DOL) and Central Population Survey (CPS) micro-level data. They also use simulated data derived from (Rothstein (2011)) that describes UI availability by state, CPS month and unemployment duration. The authors restrict their analysis to the period 2008 to 2014, to individuals aged 18-69 and to respondents who are unemployed and eligible for UI. They exploit the panel structure of CPS data to match individuals to their labor market outcomes with great precision. To analyze the impact of UI extension across the Great recession, they split the sample into two periods: the first contains data for the period 2008-2011 during which the UI benefits expanded and the second contains data for the period 2012:2014 during which UI benefits were rolled back to the normal duration. This allows them to compare the the effect over periods of benefit duration expansion and contraction. The authors identify causal estimates by using variation in EUC rules and decision by some states to cut regular benefits, a similar approach as in Rothstein (2011). The authors use a simple logistic discrete-choice model of exit from unemployment which is estimated using a logit model. Additionally, they use a competing risks version of the model to identify the separate effects, i.e. the labor force exit effect and the re-employment effect. The authors use two specifications: the first to identify the effects during the expansion and contraction phase of the extension of UI benefits and the second to identify the effects of the complete phase out of EUC and EB. I discuss the details of the specifications in the next section where I replicate the main results.

2.4 Findings

The authors document three important results. First, the data suggests that the expansion of UI duration occurred simultaneously with a sharp increase in labor market slack, intuitively, the ratio of the unemployed workers to vacancies (UV ratio). Further, the authors document that the labor market was substantially tighter (implying low UV ratio) even before the extensions began to be rolled back. Second, they observe that the total exits out

[4]Extended Benefits

[5]Emergency Unemployment Compensation

of unemployment and the exit into employment fell sharply at the epicentre of the crisis. Third, Their estimation results indicate that exits from unemployment are negatively associated with the extensions in UI, or intuitively, the higher the duration of UI benefits: the lower the rate of exits from unemployment. This is observed for both the sample periods with a magnitude of 3.5 percentage points and 2.7 percentage points respectively. The competing risks framework that they use suggest that the effect on exit from unemployment is primarily driven by the labor force exit effect; consistent with their previous work. Results from the alternative specification which captures the effect of the elimination of the EB or EUC programs are consistent with their first specification.

3 Replication of the Main Results

In this section, the results of the replication exercise are presented under two subsections: The first subsection replicates Figure 1 and Figure 2 from the original paper. The second subsection replicates the results from Table 1 of the original paper. In each subsection, I discuss additional details including data source, specifications and robustness.

The authors have made the data that they have constructed, publicly available, along with the code[6]. Due to technical constraints such as changing data-distributors over time and the sheer length of the code, I mostly use the data and code constructed by the authors to replicate the results, making modifications where there is a need to make the code consistent. Additionally, I provide scripts[7] that sets up a scraper to automate downloading of additional data that the authors do not provide, but is required in order to conduct the replication exercise.

3.1 The Expansion and Reduction of UI Benefit Durations

This section replicates Figures 1 and 2 constructed by Farber, Rothstein, and Valletta (2015). Figure 1 plots the median weeks of available UI (UI remaining), quarterly since 2007 and the UV ratio. The series for the remaining duration of unemployment benefits is constructed using the data extract provided by the authors which contains CPS micro-level data matched with the simulated data for weeks of Unemployment Insurance by state, month and unemployment duration. The series for the UV ratio, which measures the labor market slack is constructed using a series of the ratio of vacancies to unemployment, downloaded independently from the FRED[8] data base matched with the data extract provided by the authors. This series has been discontinued since 2018.

[6]Replication Materials Provided by the Authors

[7]Can be obtained from https://github.com/nikhilsebastiank/FRV2015-ReplicationMaterials

[8]Federal Reserve Bank of St. Louis, Economic Data



Figure 1: Replication of Figure 1, Weeks of UI Available and UV ratio

As we can observe from Figure 1, even though the data for the series of UV ratio was obtained independently and assigned independent weights, similar trends are observed in Figure 1 of the main analysis, except for the fact that UV ratio is obtained for one additional quarter compared to the original data set used by the authors. Hence, we observe that at the final data point, the UV ratio is slightly below 2, whereas in Figure 1. of the main analysis it is slightly above 2. This confirms the robustness of the first main observation of the paper: that the expansion of the duration of UI benefits occurred simultaneously with a sharp increase in labor market slack, but the labor markets became substantially tight, even before the additional benefits were rolled back.

Figure 2 plots the quarterly exit rates from unemployment: including the total exit rates, exits to employment and exits out of the labor force. In order to replicate Figure 2, the data extract provided by the authors was used.



Figure 2: Replication of Figure 2, Monthly Exit Rates From Unemployment

Since we have used the same data extract as used by Farber, Rothstein, and Valletta (2015), we observe the same trends as Figure 2 from the main analysis. From Figure 2, we can observe that the total exits from unemployment and exits into employment fell sharply beginning in 2008, when the impact of the crisis was substantial, confirming the replication of the second main observation of the paper.

3.2 The Effects of Extended Benefits on Exits from Unemployment

This section replicates the third important result from the analysis: the impact of extended benefits on exits from unemployment. As discussed earlier, the authors use a simple logistic discrete-choice model of exit from unemployment with competing risks to identify the separate effects, i.e. the labor force exit effect and the re-employment effect and is estimated using a logit model. The authors use two specifications of the model which are discussed below.

Specification 1

The first specification of the model is given as:

$$y^*_{ist} = X_{ist}\beta + Z_{st}\lambda + UI_{ist}\delta + \omega_s + \Psi_t + \epsilon_{ist}$$

where X_{ist} is a vector of individual variables, Z_{st} is a vector of time-varying state labor market variables, ω_s and Ψ_t are vectors of state and date effects respectively. β and λ are vectors of coefficients and ϵ_{ist} is an error term with a logistic distribution. UI_{ist} is an indicator with value = 1 if individual i, identified as unemployed in state s and at time $t-1$, will have been unemployed in month t for fewer weeks than the number of weeks of UI benefits available following the specification in Farber, Rothstein, and Valletta (2015). y^*_{ist}, the latent unobserved variable is positive if unemployment spell i ends at time t and in state s. The main parameter of interest is δ and Table 1. provides results of the estimation from the replication exercise.

Table 1: Estimated Average Marginal Effects On Probability of Exit From Unemployment, Specification 1

Variable	dy/dx	std. error	z	$P > \|z\|$	[95% conf. interval]
Single Risk 2008-11	-.0335411	.0075422	-4.45	0.000	[-.0483236, -.0187586]
Single Risk 2012-14	-.0241778	.0110373	-2.19	0.028	[-.0458105, -.0025452]
Exit to Employment 2008-11	-.0000387	.0064243	-0.01	0.995	[-.01263, .0125527]
Exit to Employment 2012-14	-.0016311	.0085041	-0.19	0.848	[-.0182988, .0150365]
Exit from Labor Force 2008-11	-.0297505	.0052667	-5.65	0.000	[-.040073, -.0194281]
Exit from Labor Force 2012-14	-.0197888	.0074906	-2.64	0.008	[-.0344701, -.0051075]

Notes:

1. Table 1 replicates the results corresponding to columns 1 and 2 of Table 1 from the main analysis.
2. It presents the estimated average marginal effects on probability of exit from unemployment, along with the standard errors, z values and 95% confidence intervals.

As we can observe, the results are very similar to those from Table 1 of the main analysis.

This confirms the robustness of the results presented in Farber, Rothstein, and Valletta (2015). Intuitively, the extension of UI benefits during the Great Recession decreased exits from unemployment, and most of this impact was driven by the labor force exit effect. Further, dividing the sample into two periods, allows for the observation of the effects during the expansion and contraction phase of the extension in UI benefits and the results are consistent with the main idea.

Specification 2

The second specification of the model is given as:

$$y^*_{ist} = X_{ist}\beta + Z_{st}\lambda + \gamma postEB_{st} + \pi_{EUC}EUCrange_{ist} + \pi_{EB}EBrange_{ist}$$
$$+\theta max\{EUCrange_{ist} \times postEUC_t, EBrange_{ist} \times postEB_{st}\} + \omega_s + \Psi_t + \mu_{ist}$$

where y^*_{ist} is the latent propensity to exit unemployment, $postEB_{st}$ and $postEUC_t$ are indicators for observations after the end of EB and EUC benefits respectively, $EUCrange_{ist}$ and $EBrange_{ist}$ are indicators for observations with unemployment durations that were covered by EUC and EB in the last month in which the relevant program was available in the state. The coefficient of interest is θ which captures the estimated marginal effects on probability of exit from unemployment.

Table 2: Estimated Average Marginal Effects On Probability of Exit From Unemployment, Specification 2

Variable	dy/dx	std. error
Single Risk 2012-14	.0158164	.018057
Exit to Employment 2012-14	.0657479	.0519364
Exit from Labor Force 2012-14	-.0038832	.0155694

Notes

1. Table 2 replicates the results corresponding to column 3 of Table 1 from the main analysis.
2. It presents the estimated average marginal effects on probability of exit from unemployment, along with the standard errors.

The main parameter of interest is θ and Table 2. presents the results of estimation of the model specified with Specification 2 from the replication exercise. We can observe some differences in the estimated results of the replication exercise with respect to column 3 of Table 1 from the main analysis. First, We can rule out the effect on the exit from labor force since it is statistically insignificant both in the main analysis and in the replication exercise. With reference to the effect on the exit to employment, there is a small effect which is weakly significant and is comparable with the main analysis. The main difference between the main analysis and the replication exercise comes from the fact that the effect on exit from unemployment becomes insignificant in the replication exercise against the

weakly significant effect observed in the main analysis. This makes it difficult to confirm the robustness of the results of the estimation of model specified by Specification 2, although, these effects are very small compared to the model in specification 1. To conclude, the replication exercise confirms the robustness of the general observations and results of the paper by Farber, Rothstein, and Valletta (2015), but questions the robustness of the results captured by the model specified with specification 2.

4 Discussion and Conclusion

The multidimensional aspect of unemployment insurance, makes it very difficult for policy-makers to determine the optimal level of unemployment insurance benefits. The analysis by Farber, Rothstein, and Valletta (2015) takes us a step ahead in answering the question of the optimal level of unemployment insurance, especially with respect to the time-dimension of UI-benefit and during periods with declining economic activity. The authors estimate the effect of extended UI benefits during the great recession on exit from unemployment and find that the extension of UI benefits is negatively associated with exit from unemployment. They document that the effect is driven primarily through the channel of labor-force attachment. This essay presents the results of a replication exercise conducted to check the robustness of the results presented by the authors. The results hold in general, but the robustness of the estimations using the model specified by specification 2 is questionable, as there seems to be some observable differences with the estimated results in the replication exercise. The authors definitely deserve credit for the methodology that they have adopted to simulate and match micro-level CPS data with the simulated data regarding the duration of unemployment insurance benefits. A significant challenge for the replication study was the fact that the distributors of CPS data have changed over time, making it difficult to obtain harmonized data. This restricts the replicator to the data extract provided by the authors. Further, to replicate the series of the UV ratio discussed in section 3.1, data had to be obtained independently from the FRED database and had to be processed. These were some of the main limitations of the replication exercise on which this essay is based.

Will the results from Farber, Rothstein, and Valletta (2015) still hold during other economic crises? For instance, the CARES[9] act was passed by the Congress in response to the COVID-19 shock to provide economic relief to all those who were affected by the pandemic in the U.S. The Pandemic Emergency Unemployment Compensation (PEUC), under the CARES act provided relief to unemployed workers, and the program was finally eliminated in September 2021. It would be very interesting to analyze the impact of the extension of benefits during the COVID-19 crisis, to better understand and extend the validity of the results in Farber, Rothstein, and Valletta (2015).

[9]https://home.treasury.gov/policy-issues/coronavirus/about-the-cares-act

References

Baily, Martin Neil (1978). "Some aspects of optimal unemployment insurance". In: *Journal of public Economics* 10.3, pp. 379–402.

Cahuc, Pierre, Stéphane Carcillo, and André Zylberberg (2014). *Labor economics*. MIT press.

Chetty, Raj (2006). "A general formula for the optimal level of social insurance". In: *Journal of Public Economics* 90.10-11, pp. 1879–1901.

Farber, Henry S, Jesse Rothstein, and Robert G Valletta (2015). "The effect of extended unemployment insurance benefits: Evidence from the 2012-2013 phase-out". In: *American Economic Review* 105.5, pp. 171–76.

Farber, Henry S and Robert G Valletta (2015). "Do extended unemployment benefits lengthen unemployment spells? Evidence from recent cycles in the US labor market". In: *Journal of Human Resources* 50.4, pp. 873–909.

Rothstein, Jesse (2011). *Unemployment insurance and job search in the Great Recession*. Tech. rep. National Bureau of Economic Research.

Smith, Sean M, Roxanna Edwards, and Hao C Duong (2021). "Unemployment rises in 2020, as the country battles the COVID-19 pandemic". In: *Monthly Lab. Rev.* 144, p. 1.

5 Appendix

A.1 : Reference Figures and Graphs from the Main Analysis



FIGURE 1. WEEKS OF UI AVAILABLE AND UNEMPLOYED/VACANCIES, QUARTERLY, 2007–2014:II

Notes: UI weeks available is median across states, weighting states by the number of UI-eligible unemployed (authors' tabulations of weighted CPS microdata). Unemployed per vacancy computed from Bureau of Labor Statistics (BLS) tabulations of CPS and JOLTS data (seasonally adjusted).

Figure 3: Figure 1 in the main analysis



FIGURE 2. MONTHLY EXIT RATES FROM UNEMPLOYMENT, 2007–2014:II (*seasonally adjusted*)

Note: Quarterly averages of monthly exit hazards, computed from weighted analysis sample of UI-eligible individuals unemployed for at least three full months.

Figure 4: Figure 2 in the main analysis

TABLE 1—ESTIMATED AVERAGE MARGINAL EFFECTS ON PROBABILITY OF EXIT FROM UNEMPLOYMENT

	Specification 1		Specification 2
Model	2008–2011	2012–2014:6	2012–2014:6
(1) Single risk	−0.034	−0.024	0.018
	(0.008)	(0.011)	(0.016)
(2) Exit to employment	−0.000	−0.002	0.018
	(0.006)	(0.009)	(0.011)
(3) Exit from labor force	−0.030	−0.020	0.001
	(0.005)	(0.007)	(0.011)

Notes: Columns 1 and 2 present the average marginal effect on the exit probability of an indicator for availability of UI benefits (a transformation of $\hat{\delta}$). Column 3 presents the average marginal effect of an indicator for the *loss* of benefits (a transformation of $\hat{\theta}$), controlling for an indicator for simulated benefit eligibility in the pre-expiration period. See text for a list of other controls included in the models and for spell counts.

Figure 5: Table 1 in the main analysis

A.2 : Code used for the Replication Exercise

The scripts can be downloaded from
https://github.com/nikhilsebastiank/FRV2015-ReplicationMaterials

Downloading additional data - Julia

```julia
# Automating the download of CPS Basic Data from NBER
# Please be careful to set the right working directory
# a - Start Year, b - End Year

function downloadCPSB(a,b)
    url = "https://data.nber.org/cps-basic2/dta/"
    k = "cpsb"
    for i in a:b
        for j in 1:12
            if j < 10
            download(url*k*string(i)*"0"*string(j)*".dta", "Data/"*string(i)*"0"*string(j)*".dta")
            else
            download(url*k*string(i)*string(j)*".dta", "Data/"*string(i)*string(j)*".dta")
            end
        end
    end
end

using CSV
# Automating the download of V-U Data from FRED to csv format:
function downloadFred()
    url = "https://fred.stlouisfed.org/graph/fredgraph.csv?bgcolor=%23e1e9f0&chart_type=line&drp=0&fo=
    download(url, "Data/VURatio.csv")
end
```

Creating Data for UV Ratio - Stata and R

```stata
* Creating the data for U-V ratio *
* This file to be loaded into the R-script*

clear all
version 13
cd /Users/nikhilsebastian/Desktop/SciencesPo/Semester3/LaborEcon/extendedUI-COVID19/ReplicationFiles/dta
import delimited fred_a.csv
gen date1 =ym(year,month)
format date1 %tm
drop if year < 2007 | year > 2014

gen yrq = qofd(dofm(date1))
format %tq yrq

save fred_a, replace
```

```
install.packages("haven")
library(haven)
library(dplyr)

# Set working directory
setwd("/Users/nikhilsebastian/Desktop/SciencesPo/Semester3/LaborEcon/extendedUI-COVID19/ReplicationFiles/dta")

# Read the cleaned dta data
data = read_dta("fred_a.dta")
data$q = 1 + (as.numeric(data$month)-1) %/% 3
View(data)

# Create a new data set with quarterly averaged unemployment to vacancy ratio
newdata = data %>%
    select(year, month, q, uvrate, yrq) %>%
    group_by(yrq) %>%
    summarize(uvrate = sum(uvrate)/3)

View(newdata)

# Export to a .dta file
write_dta(newdata, file.path(getwd(), "Newfred_a.dta"))
```

Replicating Figure 1 - Stata

```
version 13.0
capture log close
clear *
set more off
capture program drop _all
cd /Users/nikhilsebastian/Desktop/SciencesPo/Semester3/LaborEcon/extendedUI-COVID19/ReplicationFiles/dta

set linesize 120
log using logs\fig1_UI_UV_new2.log, replace

#d ;
use ext_wks ui_weeks eb_weeks eb date year month elig2 age wgtbls
  if year>=2007 & elig2==1 using ext06, clear;
drop if age<18 | age>=70; /* KEEP ONLY 18-69 */
gen yrq=qofd(dofm(date));
format %tq yrq;

collapse (mean) mean_ui_wks=ui_weeks  (p10) q10_ui_wks=ui_weeks
 (p50) med_ui_wks=ui_weeks (p90) q90_ui_wks=ui_weeks [aw=wgtbls], by(yrq);
#d cr
label var yrq "Quarter"
label var mean_ui_wks "Mean Weeks UI Available"
label var med_ui_wks "Median Weeks UI Available"
label var q10_ui_wks "10th Percentile Weeks UI Available"
merge 1:1 yrq using Newfred_a, nogen /* MERGE IN UV RATIO */
label var uvrate "Unemployed per Vacancy"
keep if yofd(dofq(yrq))>=2007

** Graph code here
set scheme s2mono
sort yrq
format yrq %tqYY!-!Qq

tsset yrq

#d ;
twoway
  (tsline med_ui_wks, c(l) lpattern(solid) msymbol(i))
  (tsline uvrate, c(l) lpattern(dash) msymbol(i) yaxis(2)),
  ysize(1.8) yscale(range(20 100) axis(1)) ylab(20(10)100, grid axis(1))
  yscale(range(1 7) axis(2)) ylab(1(1)7, axis(2))
  xscale(range(188 217)) xlabel(188(2)216, tlength(*1.8) tposition(outside) labgap(0.5))
  legend(label(1 "Median Weeks UI Available (left)") label(2 "Unemployed per Vacancy (right)")
  ring(0) pos(11) rows(2) size(medsmall) rowgap(0.4) colgap(0.4)) plotregion(margin(0))
  saving(figs\fig1_UI_UV_new2.gph, asis replace);
  graph export figs\fig1_UI_UV_new2.pdf,replace;

#d cr
log close
```

Replicating Figure 2 - Stata

```
version 13.0
capture log close
clear *
set more off
capture program drop _all
cd /Users/nikhilsebastian/Desktop/SciencesPo/Semester3/LaborEcon/extendedUI-COVID19/ReplicationFiles/dta

set linesize 120
log using logs\fig_exits_new2.log, replace

use ext06, clear
/* DO QUARTERLY */
gen yrq=qofd(dofm(date))
gen byte late=yofd(dofm(date))>=2012
format %tq yrq
drop if durmoncat2<3
keep if elig2==1
keep if age>=18 & age<=69
collapse (mean) exit2 exit2_emp exit2_nilf late [aw=wgtbls],by(yrq)

/* SEASONALLY ADJUST AND SMOOTH */
gen byte quarter=quarter(dofq(yrq))
tab quarter,gen(q)
tsset yrq
foreach x in exit2 exit2_emp exit2_nilf {
    egen m`x' = mean(`x')
    reg `x' q1-q4,hascons
    predict `x'_a,resid
    replace `x'_a=`x'_a + m`x'
    drop m`x'
        tssmooth ma `x'_as=`x'_a,window(1 1 1)
}


** Graph code here
set scheme s2mono
sort yrq
format yrq %tqYY!-!Qq
keep if yofd(dofq(yrq))>=2007

label var yrq "Quarter"
label var exit2_a "All Exits"
label var exit2_emp_a "Exits to Employment"
label var exit2_nilf_a "Exits out of Labor Force"

#d ;
graph twoway
  scatter exit2_a exit2_emp_a exit2_nilf_a yrq,
  c(l l l) ms(i i) lpattern(l .-.)
  ysize(1.8) l1("Exit Rate") yscale(range(0 0.35)) ylab(0(.05).35)
  xscale(range(188 217)) xlabel(188(2)216, tlength(*1.8) tposition(outside) labgap(0.5))
  legend(ring(0) pos(12) rows(1) size(medsmall) rowgap(0.4) colgap(1.0))
  plotregion(margin(0)) saving(figs\fig2_exits_new2.gph, asis replace);
  graph export figs\fig2_exits_new2.pdf,replace;

#d cr

log close
```

Replicating Table 1, Specification 1 - Stata

```
version 13.0
capture log close
set linesize 80
set more off
cd /Users/nikhilsebastian/Desktop/SciencesPo/Semester3/LaborEcon/extendedUI-COVID19/ReplicationFiles/dta

log using logs\reg06r2.log,replace
use ext06 if elig2==1,clear  /* ELIGIBLE */
/* DEFINE SOME MACROS */
macro define dur_FV i.durmoncat2
macro define dur_JR dur dur_sq dur_inv new_un gte26
#delimit ;
macro define X urate_st* d3lne_st* i.educ i.agecat female married fem_marr
 i.race i.majind2 i.state i.date;
macro define X2 i.educ i.agecat female married fem_marr
 nonwhite i.majind2;
#delimit cr
/* NOTE THAT RACE INCLUDES HISPANIC.  HOW WAS THIS DONE? */
/* CREATE UI AVAILABILITY VARIABLES
   NOTE THAT WKSLEFT_TOT IS MISSING FOR OBS PRIOR TO MAY 2004
   BUT STILL SOME MISSINGS BETWEEN 2004M5 AND 2006M1  */
drop if date<=ym(2004,4)
  ** USE RV VARIABLE HERE (ALREADY MATCHED TO EXIT MONTH)
gen byte ui_avail=(ui_weeks>duration2) if (ui_weeks<. & duration2<.)
label var ui_avail "=1 if weeks left >0 (in exit month)"
#delimit:
gen byte ui_last=((duration2-ui_weeks)>=-4 & (duration2-ui_weeks)<=0)
  if (ui_weeks<. & duration2<.);
#delimit cr
label var ui_last "=1 if in last 4 weeks of eligibility (in exit month)"
/* macro define ui_hf ui_avail ui_last */
macro define ui_hf ui_avail
/* DROP AGE<18 AND AGE>=70 */
tab agecat
drop if age<18 | age>=70
tab agecat
label var agecat "Decade of Age"
/* DROP IF LESS THAN 3 MONTHS UNEMPLOYMENT */
drop if durmoncat2<3
/* GENERATE STATE BY MONTH FOR CLUSTERING */
gen long mstate=(state*1000)+date
label var mstate "state*1000 + month"
/* OVERALL HAZARD */
preserve
foreach v in exit2 exit2_emp exit2_nilf {
 keep if year>=2008 & year<=2011
 di
 di "OUTCOME IS `v'"
 di
 logit `v' $ui_hf  $dur_FV $X,cluster(mstate) nolog
 margins,dydx($ui_hf)
 restore,preserve
 keep if year>=2012
 di
 di "OUTCOME IS `v'"
 di
 logit `v' $ui_hf  $dur_FV $X,cluster(mstate) nolog
 margins,dydx($ui_hf)
 restore,preserve
}
log close
set more on
```

Replicating Table 1, Specification 2 - Stata

```
version 13
cd /Users/nikhilsebastian/Desktop/SciencesPo/Semester3/LaborEcon/extendedUI-COVID19/ReplicationFiles/dta
clear *
set more off

cap log close
log using euceb_eventstudy_final.log, text replace


*Figure out which weeks are on EUC in each state at end of 2013.
*Note: I focus only on cohorts that have not had an interruption.
 use if date==mdy(12,15,2013) using stateeuc_raw, clear
 gen eucweeks=(euc1_weeks+euc2_weeks+euc3_weeks+euc4_weeks)
 keep st_cens eucweeks
 tempfile eucwks
 save `eucwks'
 use if date>=mdy(12,1,2013) & date<mdy(1,1,2014) using simulateui_cps, clear
 gen dur=(date-startdt)/7
 isid st_fips dur
 sort st_fips dur
 xtset st_fips dur
 by st_fips: gen endofregular=(wksleft_reg==0 & L.wksleft_reg>0)
 keep if endofregular==1
 gen eucstart=dur+1
 keep st_fips st_cens eucstart
 merge 1:1 st_cens using `eucwks'
 gen eucend=eucstart+eucweeks-1
 keep st_fips eucstart eucend
 save `eucwks', replace

 *Now do the same for EB, in the last month of EB in the state
  *Start by identifying that month
  *NOTE: For now, I just capture the *last* time EB went away in the state,
  *not all of them.
   use if date>mdy(1,1,2009) using stateeb_raw, clear
   sort st_cens date
   xtset st_cens date, delta(7)
   by st_cens: gen endofeb=(eb_weeks>0 & F.eb_weeks==0)
   keep if endofeb==1
   sort st_cens date
   by st_cens: keep if _n==_N
   gen day=day(date)
   gen cpsdate=date if day>=12 & day<=18
   replace cpsdate=date-day+12 + mod(day-12, 7) if day>18
   replace cpsdate=date-21 if day<12
   replace cpsdate=cpsdate+7 if day<12 & day(cpsdate)<12
   keep st_cens cpsdate eb_weeks
   rename cpsdate date
   format date %td
   tempfile ebwks
   save `ebwks'
```

```
  *Now need to figure out how many weeks of regular & EUC benefits were available then
   merge 1:1 st_cens date using stateeuc_raw, assert(2 3) keep(3)
   gen eucweeks=(euc1_weeks+euc2_weeks+euc3_weeks+euc4_weeks)
   keep st_cens date eb_weeks eucweeks
   save `ebwks', replace
   merge 1:m st_cens date using simulateui_cps, assert(2 3) keep(3)
   gen dur=(date-startdt)/7
   isid st_fips dur
   sort st_fips dur
   xtset st_fips dur
   by st_fips: gen endofregular=(wksleft_reg==0 & L.wksleft_reg>0)
   keep if endofregular==1
   gen eucstart=dur+1
   gen ebstart=dur+eucweeks+1
   gen ebend=ebstart+eb_weeks-1
  keep st_fips date ebstart ebend
  rename date ebenddate
  save `ebwks', replace

 use ext06 if elig2==1,clear  /* ELIGIBLE */
 drop if date<=ym(2004,4)
  ** USE RV VARIABLE HERE (ALREADY MATCHED TO EXIT MONTH)
 gen byte ui_avail=(ui_weeks>duration2) if (ui_weeks<. & duration2<.)
 label var ui_avail "=1 if weeks left >0 (in exit month)"

 *Assign EUCrange
 merge m:1 st_fips using `eucwks', assert(3)
 gen eucrange=(duration2>=eucstart-4 & duration2<=eucend-4)
 gen posteuc=(date>=ym(2013,12))
 *Note: The classification isn't perfect
  tab ui_avail posteuc if eucrange==1
  tab date ui_avail if eucrange==1 & date>=ym(2013,9) & date<=ym(2014,3)
 *Assign EB range
  cap drop _merge
  merge m:1 st_fips using `ebwks', assert(1 3)
  gen ebrange=(duration2>=ebstart-4 & duration2<=ebend-4)
  gen posteb=(date>=mofd(ebenddate))
  gen diff_eb=date-mofd(ebenddate)
  tab diff_eb ui_avail if ebrange==1 & diff_eb>-12 & diff_eb<12

  gen cutoff=max(eucrange*posteuc, ebrange*posteb)
  gen cutoff_eb=ebrange*posteb
  gen cutoff_euc=eucrange*posteuc

/* DEFINE SOME MACROS */
macro define dur_FV i.durmoncat2
macro define dur_JR dur dur_sq dur_inv new_un gte26
#delimit ;
macro define X urate_st* d3lne_st* i.educ i.agecat female married fem_marr
 i.race i.majind2 i.state i.date;
macro define X2 i.educ i.agecat female married fem_marr
 nonwhite i.majind2;
#delimit cr

/* DROP AGE<18 AND AGE>=70 */
tab agecat
drop if age<18 | age>=70
tab agecat
label var agecat "Decade of Age"
/* DROP IF LESS THAN 3 MONTHS UNEMPLOYMENT */
drop if durmoncat2<3
/* GENERATE STATE BY MONTH FOR CLUSTERING */
gen long mstate=(state*1000)+date
label var mstate "state*1000 + month"
```

```
** Restrict to 2012-2014m6
keep if year>=2012

table eucrange ebrange, contents(n posteb sum posteb sum cutoff)

reg ui_avail eucrange ebrange cutoff posteb $dur_FV $X, cluster(mstate)

foreach v in exit2 exit2_emp exit2_nilf {
 di
 di "SPEC 1: OUTCOME IS `v'"
 di
 logit `v' eucrange ebrange cutoff posteb $dur_FV $X,cluster(mstate) nolog
 di "SPEC 1: OUTCOME IS `v'"
 margins,dydx(eucrange ebrange cutoff posteb)

}

foreach v in exit2 exit2_emp exit2_nilf {
 di
 di "SPEC 2: OUTCOME IS `v'"
 di
 logit `v' eucrange ebrange cutoff_euc cutoff_eb posteb $dur_FV $X,cluster(mstate) nolog
 di "SPEC 2: OUTCOME IS `v'"
 margins,dydx(eucrange ebrange cutoff_euc cutoff_eb posteb)

}

log close
```